

HIRU LERROKATU



Hemen gatoz berriro ere topaleku dugun sail honetan beste programatxo bat eskaini asmoz. Dena den gauza guztietan atsedenaldira bat mesedegarri izaten dela uste dugunez, udan sartu garela eta opor-giroa urtean zehar landutako ildotik pixka bat irteteko aprobeztatu nahi izan dugu.

Orain arteko programak gehienak asiguarekin gutxi edo gehiago loturik egon badira ere (dela fisika ,dela matematika etab.ekin) ale honetan joko bat sartzea egokiago iruditu zaigu.

HIRU LERROKATU joko ezaguna izango da guzti-entzat seguruenik, baina badaezpada hemen doaz arauak.

1. Jokalari bakoitzak hiru fitxa ditu, jokalari bakoitzaren fitxak desberdinak izanik.
2. Jokoa hiru fitxa berdin lerrokatzean datza.
3. Lehenengo jokalaria erdian kokatu beharko du lehenengo fitxa eta hau ez da inoiz handik mugituko. Jokaldi bakoitzean fitxa bakarra mugituko da.

Jokalari bakoitzak lor ditzakeen soluzio posibleak hauek dira:



Programaren azalpena eta programa bera besteetan baino luzeagoak diren arren, ez beldurtu, erabilpen alde-tik oso erraza gertatuko zaizu eta. Beraz, animo eta jolastu lasai.

Programaren zatiak:

- 10-50: Programaren hasieraketa. Bertan jokalari bakoitzaren emaitza edo soluzio posibleak definitzen dira (50).
- 60-90: Taula irudikatu.
- 100-210: Jokalari 1 edo 2 izango diren aukeratu eta bakoitzaren aurkezpena egin. Jokalari bakarra baldin bada, ordenadorearekin jokatu du; jokalaria berak aukeratu beharko du joko nork hasiko duen. Jokalari bakoitzak fitxa desberdinak izango ditu; batek biribilak eta besteak koadratuak.
- 220: Lehenengo jokalaria lehen fitxa automatikoki kokatuko da erdian eta ezingo da hementik mugitu.
- 230-420: Fitxak lehen aldiz taulan kokatzea. Fitxak higitzea tokatzen zaion jokalaria ordenadorea baldin bada, azpirrutina batera

```

10  REM HIRU LERROKATU
20  KEY OFF:CLS:SCREEN 1,0:COLOR 0,1
30  DIM MAT$(9)
40  FOR I=1 TO 9:MAT$(I)="" :NEXT I
50  KAT1$="159258357456":KAT2$="123147369789"
60  FOR I=1 TO 3:LINE (96-I,24-I)-(215+I,143+I),2,B:NEXT I
70  LINE (135,24)-(136,144),2,B:LINE (176,24)-(177,144),2,B
80  LINE (96,63)-(216,64),2,B:LINE (96,103)-(216,104),2,B
90  LOCATE 4,13:PRINT "1":LOCATE 4,20:PRINT "2":LOCATE 4,27:PRINT "3":
    LOCATE 11,13:PRINT "4":LOCATE 11,27:PRINT "6":LOCATE 18,13:PRINT "7":
    LOCATE 18,20:PRINT "8":LOCATE 18,27:PRINT "9"
100 LOCATE 22,3:PRINT "Jokalari 1 edo 2: ";
110 TE$=INKEY$:WHILE TE$<>"":TE$=INKEY$:WEND
120 WHILE JK$<>"1" AND JK$<>"2":JK$=INKEY$:WEND:PRINT JK$:JK=VAL(JK$)
130 LOCATE 22,3:PRINT SPC(37)
140 IF JK=1 THEN 180
150     LOCATE 22,3:INPUT "1. jokalariaren izena: ",J1$
160     LOCATE 22,3:PRINT SPC(37)
170     LOCATE 22,3:INPUT "2. jokalariaren izena: ",J2$:GOTO 220
180 LOCATE 22,3:INPUT "Sar ezazu zure izena: ",J$
190 LOCATE 22,3:PRINT SPC(37):LOCATE 22,3:PRINT "Zuk hasi nahi al duzu (B/E)?"
200 TE$=INKEY$:WHILE TE$<>"B" AND TE$<>"b" AND TE$<>"E" AND TE$<>"e":TE$=INKEY$
    :WEND
210 IF TE$="B" OR TE$="b" THEN J1$=J$:J2$="ORD" ELSE J1$="ORD":J2$=J$
220 CIRCLE (156,84),8,1:PAINT (156,84),1,1:MAT$(5)="1"
230 FOR I=1 TO 3
240     IF I=1 THEN 340
250         IF J1$="ORD" THEN JOK=1:GOSUB 690:N=NORA:GOTO 310
260         LOCATE 22,3:PRINT SPC(37):LOCATE 23,3:PRINT SPC(37):LOCATE 22,3
265         PRINT J1$:LOCATE 23,3:PRINT "non kokatu nahi duzu (1-9): ";:
            N$=INKEY$:WHILE N$<"1" OR N$>"9" OR N$="5":N$=INKEY$:WEND:
            PRINT N$:N=VAL(N$)
270         FOR J=1 TO 400:NEXT J
280         IF MAT$(N)="" THEN 310
290         LOCATE 22,3:PRINT SPC(37):LOCATE 23,2:PRINT SPC(37):BEEP:BEEP:
            BEEP:LOCATE 22,3:PRINT "Posizio hori okupatuta dago!!!":
            FOR J=1 TO 4000:NEXT J
300         GOTO 260
310         X=116+((N-1) MOD 3)*40:Y=44+INT((N-1)/3)*40:CIRCLE (X,Y),8,1:
            PAINT (X,Y),1,1:MAT$(N)="1"
320         IF I<3 THEN 340
330         GOSUB 920
340     IF J2$="ORD" THEN JOK=0:GOSUB 690:N=NORA:GOTO 400
350     LOCATE 22,3:PRINT SPC(37):LOCATE 23,3:PRINT SPC(37):LOCATE 22,3
355     PRINT J2$:LOCATE 23,3:PRINT "non kokatu nahi duzu (1-9)? ";:
        N$=INKEY$:WHILE N$<"1" OR N$>"9" OR N$="5":N$=INKEY$:WEND:PRINT N$:
        N=VAL(N$)
360     FOR J=1 TO 400:NEXT J
370     IF MAT$(N)="" THEN 400
380     LOCATE 22,3:PRINT SPC(37):LOCATE 23,3:PRINT SPC(37):LOCATE 22,3:
        BEEP:BEEP:BEEP:PRINT "Posizio hori okupatuta dago!!!":
        FOR J=1 TO 4000:NEXT J
390     GOTO 350
400     X=116+((N-1) MOD 3)*40:Y=44+INT((N-1)/3)*40:
        LINE (X-6,Y-6)-(X+6,Y+6),1,B:PAINT (X,Y),1,1:MAT$(N)="0"
410     IF I=3 THEN GOSUB 920
420 NEXT I
430 JOK=0

```

```

440 WHILE OSATUA$="F"
450   JOK=(JOK+1) MOD 2
460   IF (JOK=1 AND J1$="ORD") OR (JOK=0 AND J2$="ORD") THEN GOSUB 690:GOTO 560
470   LOCATE 22,3:PRINT SPC(37):LOCATE 23,3:PRINT SPC(37):LOCATE 22,3:
   IF JOK=1 THEN PRINT J1$ ELSE PRINT J2$
480   LOCATE 23,3:PRINT "Nondik mugitu nahi duzu (1-9)? ";:N$=INKEY$:
   WHILE N$<"1" OR N$>"9" OR N$="5":N$=INKEY$:WEND:PRINT N$:
   NONDIK=VAL(N$)
490   IF MAT$(NONDIK)=RIGHT$(STR$(JOK),1) THEN 520
500   LOCATE 22,3:PRINT SPC(37):LOCATE 23,3:PRINT SPC(37):LOCATE 22,3:
   BEEP:BEEP:BEEP:IF MAT$(NONDIK)<>" " THEN PRINT "Bestearen fitxare";
   "kin ari zara!!!" ELSE PRINT "Hor, ez daukazu fitxarik!!!"
510   FOR J=1 TO 4000:NEXT J:GOTO 470
520   LOCATE 22,3:PRINT SPC(37):LOCATE 23,3:PRINT SPC(37):LOCATE 22,3:
   IF JOK=1 THEN PRINT J1$ ELSE PRINT J2$
530   LOCATE 23,3:PRINT " Nora mugitu nahi duzu (1-9)? ";:N$=INKEY$:
   WHILE N$<"1" OR N$>"9" OR N$="5":N$=INKEY$:WEND:PRINT N$:NORA=VAL(N$)
540   IF MAT$(NORA)=" " THEN 560
550   LOCATE 22,3:PRINT SPC(37):LOCATE 23,3:PRINT SPC(37):LOCATE 22,3:
   BEEP:BEEP:BEEP:PRINT "Posizio hori okupatuta dago!!!":
   FOR J=1 TO 4000:NEXT J:GOTO 520
560   X=116+((NONDIK-1) MOD 3)*40:Y=44+INT((NONDIK-1)/3)*40:
   IF JOK=1 THEN CIRCLE (X,Y),8,0 ELSE LINE (X-6,Y-6)-(X+6,Y+6),0,B
570   PAINT (X,Y),0,0:MAT$(NONDIK)=" "
580   X=116+((NORA-1) MOD 3)*40:Y=44+INT((NORA-1)/3)*40:
   IF JOK=1 THEN CIRCLE (X,Y),8,1 ELSE LINE (X-6,Y-6)-(X+6,Y+6),1,B
590   PAINT (X,Y),1,1:MAT$(NORA)=RIGHT$(STR$(JOK),1)
600   IF JOK=1 THEN KAT$=KAT1$:IK$="1" ELSE KAT$=KAT2$:IK$="0"
610   GOSUB 920
620 WEND
630 LOCATE 22,3:PRINT SPC(37):LOCATE 23,3:PRINT SPC(37):LOCATE 22,3:BEEP:BEEP
640 IF JK=1 THEN 670
650   IF IR$="1" THEN PRINT J1$;" !!! Irabazi duzu."
   ELSE PRINT J2$;" !!! Irabazi duzu."
660 GOTO 680
670 IF (IR$="1" AND J1$="ORD") OR (IR$="0" AND J2$="ORD")
   THEN PRINT "Irabazi dut!!!" ELSE PRINT "Irabazi duzu!!!"
680 END
690 REM JOKALDIA DEFINITU EKINTZA
700 GOSUB 920
710 IF JOK=1 THEN EGO$=BAT$:IK$="1":KAT$=KAT1$ ELSE EGO$=BI$:KAT$=KAT2$:
   IK$="0"
720 IF LEN(EGO$)>0 THEN GOSUB 1040:GOTO 740
730 NORA=1:WHILE MAT$(NORA)<>" ":NORA=NORA+1:WEND:GOTO 900
740 IF O$="E" THEN 800
750 NONDIK=0
760 FOR K=1 TO 3
770   IF K<>II AND K<>JJ THEN NONDIK=K:K=3
780 NEXT K
790 NONDIK=VAL(MID$(EGO$,NONDIK,1)):GOTO 900
800 IF JOK=1 THEN EGO$=BI$:KAT$=KAT2$:IK$="0" ELSE EGO$=BAT$:KAT$=KAT1$:
   IK$="1"
810 IF LEN(EGO$)>0 THEN GOSUB 1040:GOTO 830
820 NORA=1:WHILE MAT$(NORA)<>" ":NORA=NORA+1:WEND:GOTO 900
830 IF O$="B" THEN 870
840 IF JOK=1 THEN EGO$=BAT$:KAT$=KAT1$ ELSE EGO$=BI$:KAT$=KAT2$
850 IK$="":GOSUB 1040
860 IF NORA=0 THEN NORA=1:WHILE MAT$(NORA)<>" ":NORA=NORA+1:WEND

```

```

870 IF JOK=1 THEN EGO$=BAT$:KAT$=KAT2$:IK$="0" ELSE EGO$=BI$:KAT$=KAT1$:
      IK$="1"
880 GOSUB 1200
890 IF NONDIK=0 THEN NONDIK=1:WHILE MAT$(NONDIK)<>IK$ OR NONDIK=5
      :NONDIK=NONDIK+1:WEND
900 RETURN
910 REM AMAIA JOKALDIA DEFINITU EKINTZA
920 REM KONPROBATU EKINTZA
930 BAT$="":BI$="":OSATUA$="F"
940 FOR J=1 TO 9
950     IF MAT$(J)="1" THEN BAT$=BAT$+RIGHT$(STR$(J),1)
960     IF MAT$(J)="0" THEN BI$=BI$+RIGHT$(STR$(J),1)
970 NEXT J
980 P1=INSTR (1,KAT1$,BAT$)
990 P2=INSTR (1,KAT2$,BI$)
1000 IF P1>0 AND LEN(BAT$)=3 THEN OSATUA$="E":IR$="1"
1010 IF P2>0 AND LEN(BI$)=3 THEN OSATUA$="E":IR$="0"
1020 RETURN
1030 REM AMAIA KONPROBATU EKINTZA
1040 REM LERROA OSA DAITEKEEN BEGIRATU EKINTZA
1050 O$="E":NORA=0
1060 FOR I1=1 TO 12 STEP 3
1070     AZP$=MID$(KAT$,I1,3):KONT=0
1080     FOR J1=1 TO LEN(EGO$)
1090         XX=INSTR(1,AZP$,MID$(EGO$,J1,1))
1100         IF XX>0 THEN KONT=KONT+1:MID$(AZP$,XX,1)=" " ELSE GOTO 1130
1110         IF KONT=1 THEN II=J1
1120         IF KONT=2 THEN JJ=J1
1130     NEXT J1
1140 IF KONT<2 THEN 1160
1150 IF MAT$(VAL(AZP$))="" THEN I1=12:NORA=VAL(AZP$):O$="B":GOTO 1170
1160 IF KONT=1 AND IK$="" AND MAT$(VAL(LEFT$(AZP$,1)))="" AND
      MAT$(VAL(RIGHT$(AZP$,1)))="" THEN I1=12:IF CINT(RND)=0 THEN
      NORA=VAL(RIGHT$(AZP$,1)) ELSE NORA=VAL(LEFT$(AZP$,1))
1170 NEXT I1
1180 RETURN
1190 REM AMAIA LERROA OSA DAITEKEEN BEGIRATU EKINTZA
1200 REM NONDIK MUGITU EKINTZA
1210 NONDIK=0
1220 FOR I1=1 TO LEN(EGO$)
1230     P$=MID$(EGO$,I1,1):IF P$<>"5" THEN AURK=INSTR (1,KAT$,P$) ELSE 1330
1240     IF AURK=0 THEN 1330
1250     IF AURK MOD 3=1 THEN ALD1=VAL(MID$(KAT$,AURK+1,1)):
      ALD2=VAL(MID$(KAT$,AURK+2,1)):GOTO 1280
1260     IF AURK MOD 3=2 THEN ALD1=VAL(MID$(KAT$,AURK-1,1)):
      ALD2=VAL(MID$(KAT$,AURK+1,1)):GOTO 1280
1270     IF AURK MOD 3=0 THEN ALD1=VAL(MID$(KAT$,AURK-2,1)):
      ALD2=VAL(MID$(KAT$,AURK-1,1))
1280     IF IK$<>" " THEN 1320
1290     IF MAT$(ALD1)<>IK$ OR MAT$(ALD2)<>IK$ THEN GOTO 1330
1300     I1=3:IF ALD1<>5 THEN NORA=ALD1 ELSE NORA=ALD2
1310     GOTO 1330
1320     IF MAT$(ALD1)<>IK$ OR MAT$(ALD2)<>IK$ THEN I1=3
1330 NEXT I1
1340 NONDIK=VAL(P$)
1350 RETURN
1360 REM AMAIA NONDIK MUGITU EKINTZA

```

egingo da deia (250,340). Bertan (690) fitxa zein lekutan jartzea komeni zaion aztertuko du.

Aurrerako ikusiko dugu azterketa hori nola burutzen duen. Jokalaria ordenadora ez bada, jokalaria berari galdetuko zaio fitxa non kokatu nahi duen, bere erantzunak bete behar duen baldintza bakarra hauxe izanik: 1etik 9ra bitartekoa izatea, 5a (erdikoa) ez izatea eta posizio hori lehendik okupatuta ez egotea. Aukeratutako posizioa libre baldin badago, jokatzeko ari den jokalaria dagokion fitxa kokatuko da bertan (310, 1. jokalaria, 400, 2.arena) eta posizio hori ikur adierazgarri batez markatua geratuko da. Aukeratutako posizioa lehendik okupatuta baldin bazegoen, mezu bat emango da arazoaren berri emanez.

Jokalari batek hirugarren fitxa kokatu bezain laster, eta gerora egingo den jokaldi bakoitzaren ondoren, lerro bat osatu den ala ez egiaztatu beharko da. (330,410).

430-620: Jokalari bakoitzak hiru fitxak taulan kokatu ondoren, txandaka mugimenduak eginez jarraitu beharko dute harik eta bietako batek lerroa osatzea lortzen duen arte.

Jokalaria ordenadorea ez bada, fitxa nondik nora mugitu nahi duen galdetuko zaio eta eskatutako jokaldia logikoa baldin bada, hots, emandako bi posizioak 1etik 9ra bitartekoak baldin badira, 5a ezik (posizio hau hasieran okupatu eta ondoren ezin izango bait da ukitu), libratu nahi duen lekuan bere fitxa bat baldin bazegoen (490) eta hartu nahi duen lekua hutsik baldin bazegoen (540), programak jokaldia burutu eraziko du beste-lako egiaztapenik egin gabe (560-600). Jokaldi bakoitzaren ondoren, une horretan jokatuko duen lerroa osatzea lortu duen ala ez ikusi beharko da (610).

630-680: Programak lerro bat osatu dela detektatzen duenean, irabazlea nor izan den adieraziko du.

Orain arte ikusi dugun hau izango litzateke nolabait programa nagusia, baina askotan gertatzen den bezala, hemen ere azpirrutina programa nagusia bera bezain garrantzitsuak dira. Lehenago bat aipatu dugu; lerroa osatu den ala ez konprobatzen duena (920-1020). Horrez gain beste hiru azpirrutina gehiago daude eta hauetan definitzen da hain zuzen ere ordenadoreak jokatzerakoan kontutan hartzen dituen puntuak eta, nolabait esateko, bere "jokaera". Pasa gaiten bada azpialat hauetako bakoitza polikiago aztertzea.

920-1030: Egiaztatu ekintza: Programa nagusiak adieraziko dio zein izan den azken jokaldia egin duen jokalaria. Azpirrutina honetan jokalaria bakoitzaren fitxak okupatzen dituzten posizioak hartuko dira kontutan, bi katea desberdin sortuz; jokalaria bakoitzeko bat (BAT\$, BI\$).

Jokalaria dagokion azpikatea bere emaitza posibleen artean (KAT\$1, KAT\$2) aurkitzen den ala ez ikustea izango da azpirrutina honen eginkizuna; eta honetaz gain irabazlea zein izan den adieraztea ere bai.

690-910: Jokaldia definitu ekintza: Ordenadoreak egin behar duen jokaldia definitzen da, horretarako fitxa nondik nora mugitu behar duen kalkulatu. Fitxa nora mugitu behar duen definitzeko gehienez hiru urrats emango ditu:

1. Baldin Berak lerroa osa badezake orduan nora definitu
2. bestela Baldin besteak lerroa osa badezake orduan nora definitu
3. bestela libre dagoen lehenengo lekua definitu norratzat ambaldin

Lehenengo bi kasuetan Lerroa osa daitekeen ekintzara egiten da deia.

Nondik mugitu behar den definitzeko berriz, nondik mugitu ekintzara egingo da deia.

1040-1190: Lerroa osa daitekeen ekintza
Jokalari baten irabazteko posibilitateak aztertzen ditu, horretarako jokalaria egoera eta bere soluzio posibleak konparatuz. NORA definitzeko, egoerako bi posizioak emaitza posible baten bi posizioekin bat etorri behar dute eta emaitza osatzen duen hirugarren posizioak libre egon behar du. NORA hirugarren posizio hori izango da.

1200-1360: Nondik mugitu ekintza
Bi kasu posible bereiztu behar dira:

- 1.- Oraindik fitxaren bat kokatu gabe baldin badauka, ez du NONDIK definitu behar.
- 2.- Hiru fitxak jadanik taulan kokatuta daudenean:
 - a) Berak lerroa osa dezakeela ikusi badu, emaitzarekin bat datozen bi fitxak dauden lekuan utzi beharko ditu, bestea mugituz.
 - b) Berak lortu ezin badu, beste jokalaria eragotzi beharko dio. Horretarako bestearen posibilitateak aztertu beharko ditu eta bere fitxa bakoitza mugitzeak izan ditzakeen eraginak kontutan hartu.